// File: src/VaultManager.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import "@openzeppelin/contracts/proxy/Clones.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "./interfaces/IVaultManager.sol";
import "./interfaces/IMEVVault.sol";

/*
    Reference Materials
    https://docs.openzeppelin.com/contracts/4.x/api/proxy - openzeppelin clones

*/

contract VaultManager is Ownable, IVaultManager{

    address public MEVVaultImplementation;
    uint256 public JinoroFee;

    //This is the inverse of the information we can obtain using getAccountVault
    //The deterministic prediction can get us from user to vault, but it is also useful to store the inverse
    mapping(address => address) public vaultToAccount;

    function accountToVault(address account) public view implementationSet returns(address){
        bytes32 salt = keccak256(abi.encodePacked(account));
        return Clones.predictDeterministicAddress(MEVVaultImplementation, salt);
    }

    //If vault has been cloned into, code size will be greater than 0
    function checkIfVaultBuilt(address account) public view returns (bool) {
        uint32 size;
        address vaultAddress = accountToVault(account);
        assembly {
            size := extcodesize(vaultAddress)
        }
        return (size > 0);
    }

    //A tiny bit more expensive than calling extract directly from the vault
    //But also clones into vault address if it has not already been cloned
    function collectVaultRewards(address account) public {
        if(!checkIfVaultBuilt(account)) buildVault(account);
        IMEVVault(accountToVault(account)).extractMEV();
    }

    function collectVaultRewards() external {
        collectVaultRewards(msg.sender);
    }

    function collectVaultRewardsBulk(address[] calldata accounts) external {
        for(uint256 i = 0; i<accounts.length; i++){
            collectVaultRewards(accounts[i]);
        }
    }

    function buildVault(address account) public implementationSet returns(address) {
        bytes32 salt = keccak256(abi.encodePacked(account));

        //This will revert if the vault is already built, so no need to check that case
        address vault = Clones.cloneDeterministic(MEVVaultImplementation, salt);

        //Set starting fee inside vault
        IMEVVault(vault).updateFee();

        //populate inverse mapping
        vaultToAccount[vault] = account;
        return vault;
    }

    //This should be done by the deployment script
    function setMEVVaultImplementation(address impl) external onlyOwner {
        require(MEVVaultImplementation == address(0x0), "VAULTMANAGER: Impl Already Set");
        MEVVaultImplementation = impl;
    }

    function setJinoroFee(uint256 newFee) external onlyOwner {
        require(newFee <= 1e27, "VAULTMANAGER: Fee Above 100%");
        JinoroFee = newFee;
    }

    function takeOperationFees() external onlyOwner {
        payable(owner()).transfer(address(this).balance);
    }

    modifier implementationSet() {
        require(MEVVaultImplementation != address(0x0), "VAULTMANAGER: Impl not set"); //Keep message under 32 chars
        _;
    }

    //Allow contract to receive eth
    receive() external payable{
    }

}

// File: node_modules/@openzeppelin/contracts/proxy/Clones.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (proxy/Clones.sol)

pragma solidity ^0.8.0;

/**
 * @dev https://eips.ethereum.org/EIPS/eip-1167[EIP 1167] is a standard for
 * deploying minimal proxy contracts, also known as "clones".
 *
 * > To simply and cheaply clone contract functionality in an immutable way, this standard specifies
 * > a minimal bytecode implementation that delegates all calls to a known, fixed address.
 *
 * The library includes functions to deploy a proxy using either `create` (traditional deployment) or `create2`
 * (salted deterministic deployment). It also includes functions to predict the addresses of clones deployed using the
 * deterministic method.
 *
 * _Available since v3.4._
 */
library Clones {
    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create opcode, which should never revert.
     */
    function clone(address implementation) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            // Cleans the upper 96 bits of the `implementation` word, then packs the first 3 bytes
            // of the `implementation` address with the bytecode before the address.
            mstore(0x00, or(shr(0xe8, shl(0x60, implementation)), 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000))
            // Packs the remaining 17 bytes of `implementation` with the bytecode after the address.
            mstore(0x20, or(shl(0x78, implementation), 0x5af43d82803e903d91602b57fd5bf3))
            instance := create(0, 0x09, 0x37)
        }
        require(instance != address(0), "ERC1167: create failed");
    }

    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create2 opcode and a `salt` to deterministically deploy
     * the clone. Using the same `implementation` and `salt` multiple time will revert, since
     * the clones cannot be deployed twice at the same address.
     */
    function cloneDeterministic(address implementation, bytes32 salt) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            // Cleans the upper 96 bits of the `implementation` word, then packs the first 3 bytes
            // of the `implementation` address with the bytecode before the address.
            mstore(0x00, or(shr(0xe8, shl(0x60, implementation)), 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000))
            // Packs the remaining 17 bytes of `implementation` with the bytecode after the address.
            mstore(0x20, or(shl(0x78, implementation), 0x5af43d82803e903d91602b57fd5bf3))
            instance := create2(0, 0x09, 0x37, salt)
        }
        require(instance != address(0), "ERC1167: create2 failed");
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(
        address implementation,
        bytes32 salt,
        address deployer
    ) internal pure returns (address predicted) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(add(ptr, 0x38), deployer)
            mstore(add(ptr, 0x24), 0x5af43d82803e903d91602b57fd5bf3ff)
            mstore(add(ptr, 0x14), implementation)
            mstore(ptr, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73)
            mstore(add(ptr, 0x58), salt)
            mstore(add(ptr, 0x78), keccak256(add(ptr, 0x0c), 0x37))
            predicted := keccak256(add(ptr, 0x43), 0x55)
        }
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(
        address implementation,
        bytes32 salt
    ) internal view returns (address predicted) {
        return predictDeterministicAddress(implementation, salt, address(this));
    }
}

// File: node_modules/@openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: src/interfaces/IVaultManager.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

interface IVaultManager {

    function MEVVaultImplementation() external view returns(address);

    function JinoroFee() external view returns (uint256);

    function vaultToAccount(address user) external view returns (address);

    function accountToVault(address account) external view returns(address);

    function checkIfVaultBuilt(address account) external view returns (bool);

    function setMEVVaultImplementation(address impl) external;

    function setJinoroFee(uint256 newRate) external;

    function collectVaultRewards(address account) external;

    function collectVaultRewards() external;

    function collectVaultRewardsBulk(address[] calldata accounts) external;

    function buildVault(address account) external returns(address);

}

// File: src/interfaces/IMEVVault.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import "./IVaultManager.sol";

interface IMEVVault {

    function vaultManager() external view returns(IVaultManager);

    function lastFee() external view returns(uint256);

    function extractMEV() external;

    function updateFee() external;
}

// File: node_modules/@openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}